

# SPARQL Queries

This document contains SPARQL queries used during the example walkthrough.

## Query used to Search

**Listing 1** presents the query used to retrieve people from the VRTI graph with the full name “Samuel Beckett.

```
PREFIX crm: <http://erlangen-crm.org/current/>
PREFIX vrti: <https://ont.virtualtreasury.ie/ontology#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
SELECT DISTINCT *
WHERE {

    ?person a crm:E21_Person;
             crm:P2_has_type ?Gender .
    OPTIONAL { ?person crm:P1_is_identified_by ?First_Name .
                 ?First_Name crm:P2_has_type vrti:Forename ;
                 rdfs:label ?First_Name_Label .
             }
    OPTIONAL { ?person crm:P1_is_identified_by ?Full_Name .
                 ?Full_Name crm:P2_has_type vrti:Name ;
                 rdfs:label ?Full_Name_Label .
             }

    OPTIONAL { ?person crm:P1_is_identified_by ?Last_Name .
                 ?Last_Name crm:P2_has_type vrti:Surname ;
                 rdfs:label ?Surname_Label .
             }
    OPTIONAL { ?person crm:P1_is_identified_by ?Variant_Name .
                 ?Variant_Name crm:P2_has_type vrti:NameVariant ;
                 rdfs:label ?Variant_Name_Label .
             }
    OPTIONAL { ?person crm:P1_is_identified_by ?Alias .
                 ?Alias crm:P2_has_type vrti:SoubriquetAlias ;
                 rdfs:label ?Alias_Label .
             }

    OPTIONAL { ?person crm:P1_is_identified_by ?Patronymic_Matronymic .
                 ?Patronymic_Matronymic crm:P2_has_type vrti:PatronymicsMatronymics ;
                 rdfs:label ?Patronymic_Matronymic_Label .
             }
    OPTIONAL { ?person foaf:depiction ?Image . }
    OPTIONAL { ?person owl:sameAs ?Wikidata . }
    OPTIONAL { ?person vrti:DIB_area_of_interest ?Interest .
                 ?Interest rdfs:label ?Interest_Label . }
    OPTIONAL { ?person vrti:DIB_ID ?DIB_ID . }
    OPTIONAL { ?person crm:P71i_is_listed_in ?DIB_Page . }
    OPTIONAL { ?Source crm:P71_lists ?person . }

    OPTIONAL { ?person rdfs:seeAlso ?Related_Information_Permanent_Link . }
    OPTIONAL { ?person crm:P130_shows_features_of ?Related_Information_Non_Permanent_Link . }
    OPTIONAL { ?person vrti:associatedWith ?Associated_With . }
    OPTIONAL { ?person vrti:duplicatePerson ?Identical_To . }

    ?death a crm:E69_Death;
            crm:P93_took_out_of_existence ?person .
    OPTIONAL { ?death crm:P4_has_time-span ?Death_Date .
                 ?Death_Date crm:P81a_end_of_the_begin ?Death_Date_Lower .
                 ?Death_Date crm:P82b_end_of_the_end ?Death_Date_Upper . }
    OPTIONAL { ?death crm:P7_took_place_at ?Death_Place . }

    ?birth a crm:E67_Birth;
            crm:P98_brought_into_life ?person .
    OPTIONAL { ?birth crm:P4_has_time-span ?Birth_Date .
                 ?Birth_Date crm:P81a_end_of_the_begin ?Birth_Date_Lower .
                 ?Birth_Date crm:P82b_end_of_the_end ?Birth_Date_Upper . }
```

```

OPTIONAL { ?birth crm:P7_took_place_at ?Birth_Place . }
OPTIONAL { ?birth crm:P96_by_mother ?Mother .
          # ?Mother_Label rdfs:label|crm:P1_is_identified_by ?Mother.
          }
OPTIONAL { ?birth crm:P97_from_father ?Father .
          # ?Father rdfs:label|crm:P1_is_identified_by ?Father_Label .
          }

OPTIONAL { ?Floruit_Resource crm:P2_has_type vrti:Floruit ;
          crm:P4_has_time-span ?Floruit ;
          crm:P12_occurred_in_the_presence_of ?person .
          ?Floruit crm:P81a_end_of_the_begin ?Floruit_Lower ;
          crm:P82b_end_of_the_end ?Floruit_Upper .
          }

OPTIONAL { ?Occupation a crm:E74_Group ;
          crm:P2_has_type vrti:Occupation ;
          crm:P107_has_current_or_former_member ?person ;
          rdfs:label ?Occupation_Label . }

OPTIONAL { ?Rank a crm:E74_Group ;
          crm:P2_has_type vrti:Rank ;
          crm:P107_has_current_or_former_member ?person ;
          rdfs:label ?Rank_Label . }

OPTIONAL { ?Office a crm:E74_Group ;
          crm:P2_has_type vrti:Office ;
          crm:P107_has_current_or_former_member ?person . }

OPTIONAL { ?Organisation a crm:E74_Group ;
          crm:P2_has_type vrti:Organisation ;
          crm:P107_has_current_or_former_member ?person . }

OPTIONAL { ?person crm:note ?Comment . }
# filter query placed below
FILTER ( CONTAINS(LCASE(STR(?Full_Name_Label)), LCASE('samuel')) && CONTAINS(LCASE(STR(?Full_Name_Label)),
LCASE('beckett')) )

}
ORDER BY ?Full_Name_Label

```

**Listing 1:** Query used to retrieve people with full name “Samuel Beckett”

## Query used to Edit

**Listing 2** presents the query generated by the editor to change the occupation of Samuel Beckett from “writer” to “author”.

```

PREFIX crm: <http://erlangen-crm.org/current/>

INSERT DATA {
<https://kb.virtualtreasury.ie/occupation/IE/author> crm:P107_has_current_or_former_member <https://kb.virtualtreasury.ie/person/Beckett_Samuel-Barclay_c20_dib_a0533> .
} ;

DELETE DATA {
<https://kb.virtualtreasury.ie/occupation/IE/writer> crm:P107_has_current_or_former_member <https://kb.virtualtreasury.ie/person/Beckett_Samuel-Barclay_c20_dib_a0533> .
} ;

```

**Listing 2:** Update query generated by editor